
Créer, voir, sauvegarder et restaurer un spoule 
 
Source : http://larreya.pagesperso-orange.fr/as400/index.html 
 
Le programme exemple qui suit met en œuvre la plupart des astuces qui facilitent la vie en conception d'un état imprimé.  

 positionner en direct les caractéristiques d'un état, dont son nom  
 connaitre le numéro de ligne courant sans avoir à compter  
 afficher le spoule qui vient d'être généré, sans sortir du programme  
 sauvegarder un spoule dans un fichier  
 le restaurer en l'état, sauts de ligne compris  
  

     H            D 
     F*=================================================================== 
     F* PRTIP : Trucs et astuces d'impression 
     F* 
     F*                      Patrick LARREYA  1990-2000 
     F*=================================================================== 
     FQSYSPRT O   F      80     OA     PRINTER                        UC 
     F                                              KINFDS INFOPR 
     E                    CMD     1   7 80               Commandes 
     I* Infos impression : ligne courante. 
     IINFOPR      DS 
     I                                    B 367 3680NBLIN 
     C* Positionnement info impression 
     C                     CALL 'QCMDEXC' 
     C                     PARM CMD,1     CMDX   80 
     C                     PARM 80        CMDLEN 155 
     C*----------------------------------------------------------------- 
     C                     OPEN QSYSPRT 
     C* 
 1<--C                     EXCPTH1                                     on 
     C* Un état de test 
     C                     DO   100                        Pas boucler 
     C* L'impression dépend de la ligne en cours 
     C                     Z-ADDNBLIN     D1LIN   40 
     C                     SELEC 
     C           D1LIN     WHLE 10 
     C                     EXCPTD1 
     C           D1LIN     WHLE 20 
     C                     EXCPTD2 
     C           D1LIN     WHLE 25 
     C                     EXCPTD1 
     C                     OTHER 
     C                     EXCPTT9 
     C                     LEAVE 
     C                     ENDSL 
     C* 
     C                     ENDDO 
     C* 
     C                     CLOSEQSYSPRT 
     C* Affichage du résultat 
     C* Création du fichier de sauvegarde 
     C* Copie vers fichier de sauvegarde 
     C* Préparation de l'impression 
     C* Reconstitution du spoule 
     C* Affichage du nouveau spoule 
     C           2         DO   7         I       50 
     C                     CALL 'QCMDEXC' 
     C                     PARM CMD,I     CMDX   80 
     C                     PARM 80        CMDLEN 155 
     C                     ENDDO 
     C* 
     C                     SETON                     LR 
     C*==================================================================== 
     OQSYSPRT E   03           H1 
     O                                   50 'Test d''impression' 
     O        E 1              D1 
     O                                   20 '>>>>>>>>>>>>>>>>>>>>' 
     O                                   40 'après ligne' 

     O                         D1LIN     45 
     O                                   80 '>------------------>' 
     O        E 3              D2 
     O                                   40 '...... après ligne' 
     O                         D1LIN     45 
     O        E 2              T9 
     O                                   20 '  *** Terminé ***' 
** CMD Commandes 
OVRPRTF QSYSPRT splfname(test) USRDTA('un Test') pagesize(66 80) 
DSPSPLF  test * *LAST 
CRTPF QTEMP/CPYSPL RCDLEN(81) 
CPYSPLF TEST QTEMP/CPYSPL SPLNBR(*LAST) ctlchar(*FCFC) 
OVRPRTF QSYSPRT splfname(testB) ctlchar('*FCFC') pagesize(66 80) 
cpyf  qtemp/cpyspl qsysprt 
DSPSPLF  testB * *LAST 

 
Le source est assez simple pour ne pas demander beaucoup de commentaires. Pour une vraie application :  

 utiliser un fichier multi membres pour les sauvegardes  
 découper le travail entre du CL et du RPG  
 afficher des messages d'état en cours de traitement  
 utiliser les API pour gérer les listes de spoules.  

Si vous avez comme moi bâti des applications sans connaître CTLCHAR(*FCFC) et surtout CPYF vers QSYSPRT, certains 
programmes seront peut être à réécrire, ou pourront être complétés.  

